

Indice

1. Il quesito	2
2. Metodo del FREE-BODY	2
3. Metodo del PLV	4
4. Abbassamento dell'ordine della equazione differenziale	4
5. Metodo energetico	6
6. Integrale ellittico incompleto di prima specie	6
7. Integrale ellittico completo di prima specie	8
8. Integrale trigonometrico riconducibile all'integrale ellittico completo di prima specie	8
9. Integrazione per serie	9
10. Metodo di Runge-Kutta classico	10
11. Integrazione numerica	11
12. Algoritmo risolutivo	16
13. Esecuzione del programma	17
14. Codice dell'unità chiamante	17
15. Codice del modulo	21

Pendolo composto

1. Il quesito. E' assegnato il pendolo composto indicato in figura i cui parametri inerziali e geometrici sono

$$1.1) \begin{cases} m = 5 \text{ kg} & \leftarrow \text{massa complessiva} \\ I_G = 0.05 \text{ kg} \cdot \text{m}^2 & \leftarrow \text{momento d'inerzia baricentrico} \\ AG = 5 \text{ cm} \end{cases}$$

Si intende che I_G è calcolato rispetto all'asse baricentrale ortogonale al piano del pendolo. Si assuma che il corpo sia soggetto alla sola forza peso e che la cerniera in A_0 sia senza attrito. Allora si chiede


- 1.2) di determinare attraverso il metodo del FREE-BODY l'equazione del moto;
- 1.3) di integrarla per via numerica;
- 1.4) di ricavare le reazioni vincolari in funzione del tempo.

Si assumano le seguenti condizioni iniziali:

$$1.5) \begin{cases} \theta(t = 0) = 180^\circ \\ \dot{\theta}(t = 0) = 0 \end{cases}$$

Si osserva che un particolare pendolo composto avente le caratteristiche inerziali in 1.1 è quello la cui massa sia costituita da un cerchio di centro G , massa m e raggio dato da

$$1.6) \quad R = \sqrt{\frac{2I_G}{m}} = \sqrt{2 \frac{0.05 \text{ kg} \cdot \text{m}^2}{5 \text{ kg}}} \approx 14.2 \text{ cm}$$



Infatti per questo particolare pendolo il momento di inerzia rispetto l'asse baricentrale ortogonale al piano del moto è dato da

$$I_g = m \frac{R^2}{2} = 5 \text{ kg} \frac{(0.142 \text{ m})^2}{2} = 0.050 \text{ kg} \cdot \text{m}^2$$

ovvero proprio quello assegnato.

2. Metodo del FREE-BODY. La sollecitazione effettiva agente sul meccanismo proposto è data da

$$2.1) \begin{cases} \left. \begin{aligned} m\vec{g} &= \{0, -mg\} \\ \vec{R} &= \{R_x, R_y\} \end{aligned} \right\} \leftarrow \text{applicate in } G \\ \vec{M}_G^e = \vec{R} \times \overline{AG} = \begin{vmatrix} \hat{i} & \hat{j} & \hat{k} \\ R_x & R_y & 0 \\ AG \cos \theta & AG \sin \theta & 0 \end{vmatrix} = AG(R_x \sin \theta - R_y \cos \theta) \hat{k} \end{cases}$$

La sollecitazione d'inerzia invece è data da

$$2.2) \quad \begin{cases} -m\vec{a}_G = -m\{\ddot{x}_G, \ddot{y}_G\} \leftarrow \text{applicata in } G \\ \vec{M}_G^i = -\vec{\alpha}I_G = -\ddot{\theta}I_G\hat{k} \end{cases}$$

Essendo poi l'espressione generale del sistema fondamentale della dinamica data da

$$2.3) \quad \begin{cases} \vec{F}^e + \vec{F}^i = \vec{0} \\ \vec{M}_G^e + \vec{M}_G^i = \vec{0} \end{cases}$$

dove l'apice e si riferisce alla sollecitazione effettiva, mentre l'apice i indica la sollecitazione d'inerzia; dunque sostituendo le 2.1, 2.2 nella 2.3 abbiamo

$$2.4) \quad \begin{cases} m\vec{g} + \vec{R} - m\vec{a}_G = \vec{0} \\ \vec{R} \times \vec{AG} - \vec{\alpha}I_G = \vec{0} \end{cases} \Leftrightarrow \begin{cases} R_x - m\ddot{x}_G = 0 \\ R_y - mg - m\ddot{y}_G = 0 \\ AG(R_x \sin \theta - R_y \cos \theta) - \ddot{\theta}I_G = 0 \end{cases}$$

Si consideri ora che per le coordinate cartesiane del baricentro si ha

$$\begin{cases} x_G = AG \cos \theta \\ y_G = AG \sin \theta \end{cases} \Rightarrow \begin{cases} \dot{x}_G = -AG\dot{\theta} \sin \theta \\ \dot{y}_G = AG\dot{\theta} \cos \theta \end{cases} \Rightarrow \begin{cases} \ddot{x}_G = -AG(\ddot{\theta} \sin \theta + \dot{\theta}^2 \cos \theta) \\ \ddot{y}_G = AG(\ddot{\theta} \cos \theta - \dot{\theta}^2 \sin \theta) \end{cases}$$

Dunque le 2.4 si riscrivono

$$2.5) \quad \begin{cases} R_x + mAG(\ddot{\theta} \sin \theta + \dot{\theta}^2 \cos \theta) = 0 \\ R_y - mg - mAG(\ddot{\theta} \cos \theta - \dot{\theta}^2 \sin \theta) = 0 \\ AG(R_x \sin \theta - R_y \cos \theta) - \ddot{\theta}I_G = 0 \end{cases}$$

Dalla terza equazione si ricava

$$2.6) \quad \ddot{\theta} = \frac{AG(R_x \sin \theta - R_y \cos \theta)}{I_G}$$

Dalle prime due del 2.5 si ricavano le reazioni vincolari

$$2.7) \quad \begin{cases} R_x = -mAG(\ddot{\theta} \sin \theta + \dot{\theta}^2 \cos \theta) \\ R_y = mg + mAG(\ddot{\theta} \cos \theta - \dot{\theta}^2 \sin \theta) \end{cases}$$

Sostituendo le 2.7 nella 2.6 si ha

$$\begin{aligned} \ddot{\theta} &= mAG \frac{(-AG(\ddot{\theta} \sin \theta + \dot{\theta}^2 \cos \theta) \sin \theta - (g + AG(\ddot{\theta} \cos \theta - \dot{\theta}^2 \sin \theta)) \cos \theta)}{I_G} \Leftrightarrow \\ \Leftrightarrow \ddot{\theta} &= mAG \frac{(-AG\ddot{\theta} \sin^2 \theta - AG\dot{\theta}^2 \sin \theta \cos \theta - g \cos \theta - AG\ddot{\theta} \cos^2 \theta + AG\dot{\theta}^2 \sin \theta \cos \theta)}{I_G} \Leftrightarrow \\ &\Leftrightarrow \ddot{\theta} = -mAG \frac{AG\ddot{\theta} + g \cos \theta}{I_G} \Leftrightarrow \ddot{\theta}I_G = -mAG^2\ddot{\theta} - mAGg \cos \theta \end{aligned}$$

Abbiamo trovato dunque la legge del moto

$$2.8) \quad (I_G + AG^2m)\ddot{\theta} + AGmg \cos \theta = 0 \Leftrightarrow \ddot{\theta} = -\frac{1}{2} \frac{2AGmg}{I_G + AG^2m} \cos \theta = -\frac{1}{2A^2} \cos \theta$$

Le 2.7 risolvono il quesito 1.4, la 2.8 risponde al quesito 1.2.

3. Metodo del PLV. Come verifica della 2.8 vediamo ora di risolvere il quesito 1.2 attraverso l'applicazione del Principio dei Lavori Virtuali; il lavoro virtuale della sollecitazione effettiva (attiva e reattiva) e della sollecitazione inerziale si scrive

$$3.1) \quad \delta L = \vec{R} \cdot \vec{\delta}(A) + (m\vec{g} - m\vec{a}_G) \cdot \vec{\delta}(G) - \vec{\alpha} \cdot \delta\theta \hat{k} I_G$$

Si consideri ora che $\vec{\delta}(A) = 0$ per rispetto del vincolo di cerniera; inoltre si ha

$$3.2) \quad \vec{\delta}(G) = \begin{Bmatrix} \dot{x}_G dt \\ \dot{y}_G dt \end{Bmatrix} = \begin{Bmatrix} -AG\dot{\theta} dt \sin \theta \\ AG\dot{\theta} dt \cos \theta \end{Bmatrix} = \begin{Bmatrix} -AG \sin \theta \delta\theta \\ AG \cos \theta \delta\theta \end{Bmatrix}$$

Quindi la 3.1 si scrive

$$3.3) \quad \delta L = m \left(\begin{Bmatrix} 0 \\ -g \end{Bmatrix} - \begin{Bmatrix} -AG(\ddot{\theta} \sin \theta + \dot{\theta}^2 \cos \theta) \\ AG(\ddot{\theta} \cos \theta - \dot{\theta}^2 \sin \theta) \end{Bmatrix} \right) \cdot \begin{Bmatrix} -AG \sin \theta \delta\theta \\ AG \cos \theta \delta\theta \end{Bmatrix} - \ddot{\theta} \delta\theta I_G$$

Il PLV porge poi $\delta L = 0$ ovvero

$$\begin{aligned} & m \left(\begin{Bmatrix} 0 \\ -g \end{Bmatrix} - \begin{Bmatrix} -AG(\ddot{\theta} \sin \theta + \dot{\theta}^2 \cos \theta) \\ AG(\ddot{\theta} \cos \theta - \dot{\theta}^2 \sin \theta) \end{Bmatrix} \right) \cdot \begin{Bmatrix} -AG \sin \theta \\ AG \cos \theta \end{Bmatrix} - \ddot{\theta} I_G = 0 \Leftrightarrow \\ \Leftrightarrow & m \left(\begin{Bmatrix} 0 \\ -g \end{Bmatrix} \cdot \begin{Bmatrix} -AG \sin \theta \\ AG \cos \theta \end{Bmatrix} - \begin{Bmatrix} -AG(\ddot{\theta} \sin \theta + \dot{\theta}^2 \cos \theta) \\ AG(\ddot{\theta} \cos \theta - \dot{\theta}^2 \sin \theta) \end{Bmatrix} \cdot \begin{Bmatrix} -AG \sin \theta \\ AG \cos \theta \end{Bmatrix} \right) - \ddot{\theta} I_G = 0 \Leftrightarrow \\ \Leftrightarrow & m \left(-gAG \cos \theta - AG^2 \left((\ddot{\theta} \sin \theta + \dot{\theta}^2 \cos \theta) \sin \theta + (\ddot{\theta} \cos \theta - \dot{\theta}^2 \sin \theta) \cos \theta \right) \right) = \ddot{\theta} I_G \Leftrightarrow \\ \Leftrightarrow & m \left(-gAG \cos \theta - AG^2(\ddot{\theta} \sin^2 \theta + \ddot{\theta} \cos^2 \theta) \right) = \ddot{\theta} I_G \Leftrightarrow m(-gAG \cos \theta - AG^2\ddot{\theta}) = \ddot{\theta} I_G \end{aligned}$$

Abbiamo trovato dunque

$$3.4) \quad AGmg \cos \theta + (mAG^2 + I_G)\ddot{\theta} = 0$$

Si ritrova così la 2.8.

4. Abbassamento dell'ordine della equazione differenziale. Nella 3.4 si ponga

$$4.1) \quad z(\theta) = \dot{\theta}(t)$$

Allora si ha

$$4.2) \quad \ddot{\theta}(t) = \frac{dz(\theta)}{d\theta} \dot{\theta}(t)$$

Sostituendo le 4.1, 4.2 nella 3.4 si ha

$$4.3) \quad AGmg \cos \theta + (mAG^2 + I_G) \frac{dz(\theta)}{d\theta} z(\theta) = 0$$

Questa equazione differenziale si può agevolmente risolvere per separazione di variabili, infatti

$$\frac{dz(\theta)}{d\theta} z(\theta) = -\frac{AGmg \cos \theta}{mAG^2 + I_G} \Leftrightarrow z dz = -\frac{AGmg}{mAG^2 + I_G} \cos \theta d\theta \Leftrightarrow \frac{z^2}{2} = -\frac{AGmg}{mAG^2 + I_G} \int d \sin \theta$$

Quindi si trova

$$4.4) \quad z^2 = -2 \frac{AGmg}{mAG^2 + I_G} \sin \theta + C$$

Sostituendo poi la 4.1 nella 4.4 si ha

$$\dot{\theta}^2(t) = -2 \frac{AGmg}{mAG^2 + I_G} \sin \theta(t) + C$$

Imponendo poi le condizioni iniziali si ha

$$0 = -2 \frac{AGmg}{mAG^2 + I_G} \sin \theta_0 + C \Rightarrow C = 2 \frac{AGmg}{mAG^2 + I_G} \sin \theta_0$$

e quindi si conclude

$$4.5) \quad \dot{\theta}^2(t) = 2 \frac{AGmg}{mAG^2 + I_G} [\sin \theta_0 - \sin \theta(t)]$$

Si osservi ora che la 4.5 impone che sia sempre $\sin \theta_0 \geq \sin \theta(t)$, ovvero che il baricentro del corpo non risalga mai oltre la posizione assunta all'istante iniziale. Questo è perfettamente compatibile con la conservazione dell'energia del sistema.

Per risolvere l'equazione differenziale dobbiamo ora metterla in forma normale, ovvero occorre estrarre la radice; questo limiterà la validità della soluzione all'ambito temporale in cui la velocità non cambia di segno. Poiché a partire dalle condizioni iniziali 1.5 la velocità angolare risulta positiva, segue che sceglieremo di scrivere la 4.5 nella forma seguente

$$4.6) \quad \begin{cases} \dot{\theta}(t) = \frac{1}{A} \sqrt{\sin \theta_0 - \sin \theta(t)} \\ \frac{1}{A} = \sqrt{2 \frac{AGmg}{mAG^2 + I_G}} = 8.8588s^{-1} \\ \theta \in [\pi, 2\pi] \end{cases}$$

Risolvo ora la 4.6 ancora per separazione di variabili:

$$\frac{d\theta(t)}{dt} = \frac{1}{A} \sqrt{\sin \theta_0 - \sin \theta(t)} \Leftrightarrow A \frac{d\theta}{\sqrt{\sin \theta_0 - \sin \theta(t)}} = dt \Leftrightarrow$$

$$4.7) \quad t = A \int_{\theta_0}^{\theta(t)} \frac{d\gamma}{\sqrt{\sin \theta_0 - \sin \gamma}}, \quad \theta \in \left[\theta_0, \frac{3\pi}{2} \right]$$

L'integrale a secondo membro non può essere risolto in forma esatta, si tratta cioè di un integrale ellittico.

5. Metodo energetico. Provo ora come sia possibile pervenire alla 4.6 molto velocemente con considerazioni energetiche. L'energia cinetica del pendolo composto è data da

$$5.1) \quad T = \frac{1}{2}mv_G^2 + \frac{1}{2}I_G\dot{\theta}^2 = \frac{1}{2}m(v_G^2) + \frac{1}{2}I_G\dot{\theta}^2 = \left(\frac{1}{2}mAG^2 + \frac{1}{2}I_G\right)\dot{\theta}^2$$

L'energia potenziale associata al campo gravitazionale è data invece da

$$\begin{aligned} \Pi = -L &= - \int m\vec{g} \cdot \vec{v}_G dt = -m \int \begin{Bmatrix} 0 \\ -g \end{Bmatrix} \cdot \begin{Bmatrix} \dot{x}_G \\ \dot{y}_G \end{Bmatrix} dt = -m \int \begin{Bmatrix} 0 \\ -g \end{Bmatrix} \cdot \begin{Bmatrix} -AG\dot{\theta} \sin \theta \\ AG\dot{\theta} \cos \theta \end{Bmatrix} dt = \\ &= mgAG \int \dot{\theta} \cos \theta dt = mgAG \int \cos \theta d\theta \end{aligned}$$

Dunque abbiamo

$$5.2) \quad \Pi = mgAG \sin \theta + C$$

L'energia totale del sistema è pertanto

$$5.3) \quad E = T + \Pi = \left(\frac{1}{2}mAG^2 + \frac{1}{2}I_G\right)\dot{\theta}^2 + mgAG \sin \theta + C$$

D'altra parte l'energia totale relativa alle condizioni iniziali 1.5 è tutta potenziale e vale

$$5.4) \quad E = \Pi(t_0) = mgAG \sin \theta_0 + C$$

Sostituendo la 5.4 nella 5.3 abbiamo

$$mgAG \sin \theta_0 + C = \left(\frac{1}{2}mAG^2 + \frac{1}{2}I_G\right)\dot{\theta}^2 + mgAG \sin \theta + C \Rightarrow$$

$$5.5) \quad \left(\frac{1}{2}mAG^2 + \frac{1}{2}I_G\right)\dot{\theta}^2 + mgAG(\sin \theta - \sin \theta_0) = 0$$

che è proprio la 4.6.

6. Integrale ellittico incompleto di prima specie. In questo paragrafo discutiamo un particolare integrale -detto ellittico di prima specie- a cui ricondurrò in un secondo momento quello in 4.7. Si tratta del seguente integrale

$$6.1) \quad \begin{cases} F_1(k, \phi) = \int_0^\phi \frac{d\varphi}{\sqrt{1-k^2 \sin^2 \varphi}} \\ k \sin \varphi \in]-1,1[\end{cases}$$

il quale non può essere risolto in forma esatta. Qui ne propongo la soluzione per mezzo di uno sviluppo in serie di potenze. A tale scopo si consideri il noto sviluppo

$$\left\{ \begin{aligned} (1+x)^\alpha &= \sum_{k=0}^{\infty} \binom{\alpha}{k} x^k = \sum_{k=0}^{\infty} \frac{\alpha!}{k!(\alpha-k)!} x^k = 1 + \frac{\alpha!}{(\alpha-1)!} x + \frac{\alpha!}{2!(\alpha-2)!} x^2 + \frac{\alpha!}{3!(\alpha-3)!} x^3 + \dots \\ &\quad \forall \alpha \in \mathbb{R}, \forall x \in]-1, 1[\end{aligned} \right.$$

Operando le sostituzioni $x = -k^2 \sin^2 \varphi$ e $\alpha = -1/2$ abbiamo

$$\begin{aligned} \frac{1}{\sqrt{1-k^2 \sin^2 \varphi}} &= \sum_{k=0}^{\infty} \frac{(-1/2)!}{k!(-1/2-k)!} (-k^2 \sin^2 \varphi)^k = \\ &= 1 + \frac{1}{2} k^2 \sin^2 \varphi + \frac{\left(-\frac{1}{2}\right)\left(-\frac{1}{2}-1\right)}{2} k^4 \sin^4 \varphi - \frac{\left(-\frac{1}{2}\right)\left(-\frac{1}{2}-1\right)\left(-\frac{1}{2}-2\right)}{6} k^6 \sin^6 \varphi + \\ &\quad + \frac{\left(-\frac{1}{2}\right)\left(-\frac{1}{2}-1\right)\left(-\frac{1}{2}-2\right)\left(-\frac{1}{2}-3\right)}{24} k^8 \sin^8 \varphi + \dots = \\ &= 1 + \frac{1}{2} k^2 \sin^2 \varphi + \frac{3}{8} k^4 \sin^4 \varphi + \frac{15}{48} k^6 \sin^6 \varphi + \frac{105}{384} k^8 \sin^8 \varphi + \dots \end{aligned}$$

Dunque l'integrale $F_1(k, \phi)$ può essere scritto in modo equivalente come

$$6.2) \quad F_1(k, \phi) = \int_0^\phi \sum_{k=0}^{\infty} \frac{(-1/2)!}{k!(-1/2-k)!} (k^2 \sin^2 \varphi)^k d\varphi$$

Se in particolare ci limitiamo al quinto addendo dello sviluppo in serie, allora una scrittura approssimata di $F_1(k, \phi)$ è la seguente

$$\begin{aligned} F_1(k, \phi) &\cong \int_0^\phi \left(1 + \frac{1}{2} k^2 \sin^2 \varphi + \frac{3}{8} k^4 \sin^4 \varphi + \frac{15}{48} k^6 \sin^6 \varphi + \frac{105}{384} k^8 \sin^8 \varphi \right) d\varphi = \\ &= \phi + \frac{1}{2} k^2 \int_0^\phi \sin^2 \varphi d\varphi + \frac{3}{8} k^4 \int_0^\phi \sin^4 \varphi d\varphi + \frac{15}{48} k^6 \int_0^\phi \sin^6 \varphi d\varphi + \frac{105}{384} k^8 \int_0^\phi \sin^8 \varphi d\varphi \end{aligned}$$

Per gli integrali trigonometrici si calcola poi

$$\begin{aligned} \int_0^\phi \sin^2 \varphi d\varphi &= \frac{\varphi - \sin \varphi \cos \varphi}{2} \Big|_0^\phi = \frac{\phi - \sin \phi \cos \phi}{2} \\ \int_0^\phi \sin^4 \varphi d\varphi &= \frac{3\varphi - 3 \sin \varphi \cos \varphi - 2 \sin^3 \varphi \cos \varphi}{8} \Big|_0^\phi = \frac{3\phi - 3 \sin \phi \cos \phi - 2 \sin^3 \phi \cos \phi}{8} \\ \int_0^\phi \sin^6 \varphi d\varphi &= \left(-\frac{\sin^5 \varphi \cos \varphi}{6} + \frac{15}{48} (\varphi - \sin \varphi \cos \varphi) - \frac{5}{24} \sin^3 \varphi \cos \varphi \right) \Big|_0^\phi = \\ &= -\frac{\sin^5 \phi \cos \phi}{6} + \frac{15}{48} (\phi - \sin \phi \cos \phi) - \frac{5}{24} \sin^3 \phi \cos \phi \\ \int_0^\phi \sin^8 \varphi d\varphi &= \frac{7}{8} \int_0^\phi \sin^6 \varphi d\varphi - \frac{\sin^7 \varphi \cos \varphi}{8} \Big|_0^\phi = \\ &= -7 \frac{\sin^5 \phi \cos \phi}{48} + \frac{105}{384} (\phi - \sin \phi \cos \phi) - \frac{35}{192} \sin^3 \phi \cos \phi - \frac{\sin^7 \phi \cos \phi}{8} \end{aligned}$$

Dunque l'integrale F_1 può scriversi in maniera approssimata come

$$\begin{aligned}
F_1(k, \phi) \cong & \phi + \frac{1}{2}k^2 \frac{\phi - \sin \phi \cos \phi}{2} + \frac{3}{8}k^4 \frac{3\phi - 3 \sin \phi \cos \phi - 2 \sin^3 \phi \cos \phi}{8} + \\
& + \frac{15}{48}k^6 \left(-\frac{\sin^5 \phi \cos \phi}{6} + \frac{15}{48}(\phi - \sin \phi \cos \phi) - \frac{5}{24} \sin^3 \phi \cos \phi \right) + \\
& + \frac{105}{384}k^8 \left(-7\frac{\sin^5 \phi \cos \phi}{48} + \frac{105}{384}(\phi - \sin \phi \cos \phi) - \frac{35}{192} \sin^3 \phi \cos \phi - \frac{\sin^7 \phi \cos \phi}{8} \right)
\end{aligned}$$

7. Integrale ellittico completo di prima specie. Quando si ponga $\phi = \pi/2$ nell'integrale ellittico incompleto di prima specie, allora si ottiene la forma completa, ovvero l'integrale

$$7.1) \quad \begin{cases} K_1(k) = \int_0^{\pi/2} \frac{d\varphi}{\sqrt{1-k^2 \sin^2 \varphi}} \\ k \sin \varphi \in]-1,1[\end{cases}$$

Operando la sostituzione $\phi = \pi/2$ nella espressione approssimata ottenuta per la forma incompleta abbiamo

$$\begin{aligned}
K_1(k) \cong & \frac{\pi}{2} + \frac{\pi}{8}k^2 + \frac{9}{128}\pi k^4 + \frac{225}{4608}\pi k^6 + \frac{11025}{294912}\pi k^8 \cong \\
\cong & 1.5707 + 0.3926k^2 + 0.2208k^4 + 0.1533k^6 + 0.1174k^8
\end{aligned}$$

Dunque possiamo scrivere

$$7.2) \quad K_1(k) \cong 1.5707 + 0.3926k^2 + 0.2208k^4 + 0.1533k^6 + 0.1174k^8$$

8. Integrale trigonometrico riconducibile a integrale ellittico completo di prima specie. Si consideri l'integrale

$$8.1) \quad \int_0^\theta \frac{d\gamma}{\sqrt{\cos \gamma - \cos \theta}}, \quad \theta \in [0, \pm\pi]$$

dove $i = 0,1,2 \dots$ Si operino le sostituzioni seguenti

$$\begin{cases} \sin \frac{\theta}{2} = k \\ \sin \frac{\gamma}{2} = k \sin \varphi \end{cases}$$

Allora abbiamo che

$$\begin{cases} \cos \gamma = \cos^2 \frac{\gamma}{2} - \sin^2 \frac{\gamma}{2} = 1 - 2 \sin^2 \frac{\gamma}{2} \\ \cos \theta = \cos^2 \frac{\theta}{2} - \sin^2 \frac{\theta}{2} = 1 - 2 \sin^2 \frac{\theta}{2} \end{cases} \Rightarrow \sqrt{\cos \gamma - \cos \theta} = \sqrt{2} \sqrt{\sin^2 \frac{\theta}{2} - \sin^2 \frac{\gamma}{2}}$$

$$\begin{cases} \gamma = \theta \Rightarrow \begin{cases} \sin \frac{\theta}{2} = k \sin \phi \\ \sin \frac{\theta}{2} = k \end{cases} \Rightarrow \sin \phi = 1 \Rightarrow \phi = \frac{\pi}{2} \\ \gamma = 0 \Rightarrow 0 = k \sin \phi \Rightarrow \phi = 0 \end{cases}$$

$$d\left(\sin \frac{\gamma}{2}\right) = d(k \sin \phi) \Rightarrow \frac{1}{2} \cos \frac{\gamma}{2} d\gamma = k \cos \phi d\phi \Rightarrow d\gamma = \frac{2k \cos \phi}{\cos \frac{\gamma}{2}} d\phi$$

Dunque la sostituzione proposta porge

$$\int_0^\theta \frac{d\gamma}{\sqrt{\cos \gamma - \cos \theta}} = \int_0^{\frac{\pi}{2}} \frac{2k \cos \phi}{\sqrt{2} \cos \frac{\gamma}{2} \sqrt{\sin^2 \frac{\theta}{2} - \sin^2 \frac{\gamma}{2}}} d\phi = \sqrt{2} \int_0^{\frac{\pi}{2}} \frac{k \cos \phi}{\cos \frac{\gamma}{2} |k| \sqrt{1 - \sin^2 \phi}} d\phi$$

Si consideri ora che poiché $\phi \in [0, \frac{\pi}{2}]$ si ha $\cos \phi \geq 0$ e dunque $\sqrt{1 - \sin^2 \phi} = \cos \phi$, pertanto abbiamo ottenuto sin qui

$$\int_0^\theta \frac{d\gamma}{\sqrt{\cos \gamma - \cos \theta}} = \sqrt{2} \int_0^{\frac{\pi}{2}} \frac{k}{\cos \frac{\gamma}{2} |k|} d\phi$$

A questo punto si deve specificare l'ambito di variabilità di θ . Si hanno solo i seguenti casi:

$$\begin{aligned} \theta \in [0, \pi] &\Rightarrow \begin{cases} k \geq 0 \\ \cos \frac{\gamma}{2} \geq 0 \end{cases} \Rightarrow \int_0^\theta \frac{d\gamma}{\sqrt{\cos \gamma - \cos \theta}} = \sqrt{2} \int_0^{\frac{\pi}{2}} \frac{1}{\sqrt{1 - k^2 \sin^2 \phi}} d\phi \\ \theta \in [0, -\pi] &\Rightarrow \begin{cases} k \leq 0 \\ \cos \frac{\gamma}{2} \geq 0 \end{cases} \Rightarrow \int_0^\theta \frac{d\gamma}{\sqrt{\cos \gamma - \cos \theta}} = -\sqrt{2} \int_0^{\frac{\pi}{2}} \frac{1}{\sqrt{1 - k^2 \sin^2 \phi}} d\phi \end{aligned}$$

Si riconosce a secondo membro l'integrale ellittico di prima specie in forma completa, dunque possiamo fornire la seguente soluzione approssimata:

$$\begin{aligned} 8.2) \quad \int_0^\theta \frac{d\gamma}{\sqrt{\cos \gamma - \cos \theta}} &= \pm \sqrt{2} K_1 \left(\sin \frac{\theta}{2} \right) \cong \\ &\cong \pm \sqrt{2} \left[1.5707 + 0.3926 \left(\sin \frac{\theta}{2} \right)^2 + 0.2208 \left(\sin \frac{\theta}{2} \right)^4 + 0.1533 \left(\sin \frac{\theta}{2} \right)^6 + 0.1174 \left(\sin \frac{\theta}{2} \right)^8 \right] \end{aligned}$$

dove $\theta \in [0, \pm\pi]$.

9. Integrazione per serie. Vediamo ora come ricondurre l'integrale in 4.7 a quello discusso nel precedente paragrafo. Operando la sostituzione $\theta = \alpha + 3\pi/2$ nell'integrale 4.7 abbiamo

$$9.1) \quad \begin{cases} \sin \theta_0 = \sin \left(\alpha_0 + \frac{3\pi}{2} \right) = -\cos \alpha_0 \\ \sin \theta = \sin \left(\alpha + \frac{3\pi}{2} \right) = -\cos \alpha \\ \theta_0 = \alpha_0 + \frac{3\pi}{2}, \theta = \alpha + \frac{3\pi}{2}, d\theta = d\alpha \end{cases} \Rightarrow \int_{\theta_0}^{\theta(t)} \frac{d\gamma}{\sqrt{\sin \theta_0 - \sin \gamma}} = \int_{\alpha_0}^{\alpha(t)} \frac{d\beta}{\sqrt{\cos \beta - \cos \alpha_0}}$$

Possiamo scrivere anche

$$9.2) \quad \int_{\theta_0}^{\theta(t)} \frac{d\gamma}{\sqrt{\sin \theta_0 - \sin \gamma}} = - \int_0^{\alpha_0} \frac{d\beta}{\sqrt{\cos \beta - \cos \alpha_0}} + \int_0^{\alpha(t)} \frac{d\beta}{\sqrt{\cos \beta - \cos \alpha_0}}$$

Considerando i risultati del precedente paragrafo -e tendo conto delle condizioni iniziali in 1.5- si ha

$$9.3) \quad \begin{cases} \int_0^{-\frac{\pi}{2}} \frac{d\beta}{\sqrt{\cos \beta - \cos \alpha_0}} = -\sqrt{2}K_1 \left(\sin \frac{\pi}{4} \right) \\ \int_0^{\alpha(t)} \frac{d\beta}{\sqrt{\cos \beta - \cos \alpha_0}} = -\sqrt{2}K_1 \left(\sin \frac{\alpha}{2} \right) \\ \alpha \in \left[-\frac{\pi}{2}, \theta - \frac{3\pi}{2} \right] \subseteq \left[-\frac{\pi}{2}, 0 \right] \end{cases}$$

Dunque l'equazione 4.7 si può riscrivere

$$9.4) \quad t = A\sqrt{2} \left(K_1 \left(\sin \frac{\pi}{4} \right) - K_1 \left(\sin \frac{\alpha}{2} \right) \right), \quad \theta \in \left[\pi, \frac{3\pi}{2} \right], \alpha = \theta - \frac{3\pi}{2}$$

dove si deve sostituire

$$9.5) \quad \begin{cases} K_1 \left(\sin \frac{\pi}{4} \right) \cong \left[1.5707 + 0.3926 \left(\sin \frac{\pi}{4} \right)^2 + 0.2208 \left(\sin \frac{\pi}{4} \right)^4 + 0.1533 \left(\sin \frac{\pi}{4} \right)^6 + 0.1174 \left(\sin \frac{\pi}{4} \right)^8 \right] \cong 1.8450 \\ K_1 \left(\sin \frac{\alpha}{2} \right) \cong \left[1.5707 + 0.3926 \left(\sin \frac{\alpha}{2} \right)^2 + 0.2208 \left(\sin \frac{\alpha}{2} \right)^4 + 0.1533 \left(\sin \frac{\alpha}{2} \right)^6 + 0.1174 \left(\sin \frac{\alpha}{2} \right)^8 \right] \end{cases}$$

In realtà si trova che le 9.5, 9.6 non funzionano, e non so bene perché. Invece la 9.3 funziona se la si usa per calcolare il tempo necessario a descrivere un quarto della oscillazione. In tal caso infatti si ottiene

$$\int_{\pi}^{\frac{3}{2}\pi} \frac{d\gamma}{\sqrt{\sin \theta_0 - \sin \gamma}} = - \int_0^{-\frac{\pi}{2}} \frac{d\beta}{\sqrt{\cos \beta - \cos \alpha_0}} = \sqrt{2}K_1 \left(\sin \frac{\pi}{4} \right) \Rightarrow t = A\sqrt{2}K_1 \left(\sin \frac{\pi}{4} \right) \cong 0.2951s$$

Dunque si ottiene un periodo

$$9.6) \quad T = 4 \cdot 0.2951s = 1.18s$$

10. Metodo di Runge-Kutta classico. Ricordo il metodo di Runge-Kutta classico per l'integrazione numerica, secondo il quale assegnato il problema di Cauchy

$$10.1) \quad \begin{cases} \frac{dy(t)}{dx} = f(x, y(t)) \\ y(t_0) = y_0 \end{cases}$$

con un'unica soluzione in $I = [t_0, t_0 + \beta]$, allora definita la partizione seguente di I in $n + 1$ nodi

$$10.2) \quad x_i = x_0 + i \frac{\beta}{n} = x_0 + ih, \quad i = 0, 1, 2, \dots, n$$

segue che la funzione incognita può essere costruita secondo l'espressione

$$10.3) \quad y_{i+1} = y_i + \frac{h}{6} (k_1^i + 2k_2^i + 2k_3^i + k_4^i)$$

con

$$10.4) \quad \begin{cases} k_1^i = f(x_i, y_i) \\ k_2^i = f\left(x_i + \frac{h}{2}, y_i + \frac{h}{2}k_1^i\right) \\ k_3^i = f\left(x_i + \frac{h}{2}, y_i + \frac{h}{2}k_2^i\right) \\ k_4^i = f(x_i + h, y_i + hk_3^i) \end{cases}$$

11. Integrazione numerica. Volendo applicare il metodo di integrazione esposto nel precedente paragrafo al problema di Cauchy qui in esame, che ricordo

$$11.1) \quad \begin{cases} \dot{\theta}(t) = \frac{1}{A} \sqrt{-\sin \theta(t)} & \text{con } A = \frac{1}{\sqrt{2 \frac{AGmg}{mAG^2 + IG}}} \\ \theta(t=0) = 180^\circ \\ \dot{\theta}(t=0) = 0 \\ \theta \in [\pi, 2\pi] \end{cases}$$

si utilizza quanto esposto nel precedente paragrafo. In particolare, si definisce la partizione

$$11.2) \quad t_i = x_0 + i \frac{\beta}{n} = x_0 + ih, \quad i = 0, 1, 2, \dots, n$$

Quindi il metodo Runge-Kutta classico porge l'algoritmo

$$11.3) \quad \theta_{i+1} = \theta_i + \frac{h}{6} (k_1^i + 2k_2^i + 2k_3^i + k_4^i)$$

con


$$11.4) \quad \begin{cases} k_1^i = \frac{1}{A} \sqrt{-\sin \theta_i} \\ k_2^i = \frac{1}{A} \sqrt{-\sin\left(\theta_i + \frac{h}{2}k_1^i\right)} \\ k_3^i = \frac{1}{A} \sqrt{-\sin\left(\theta_i + \frac{h}{2}k_2^i\right)} \\ k_4^i = \frac{1}{A} \sqrt{-\sin(\theta_i + hk_3^i)} \end{cases}$$


Una volta che si sia ricavato dalla 10.3 l'array dei valori di θ è possibile ricavare dalla 3.4 quello dei corrispondenti valori di $\ddot{\theta}$ e l'array dei valori di $\dot{\theta}_i$ dalla prima delle 4.6, avendosi


$$11.5) \begin{cases} \dot{\theta}_i = \frac{1}{A} \sqrt{-\sin \theta_i} \\ \ddot{\theta}_i = -\frac{AGmg}{mAG^2 + I_G} \cos \theta_i \end{cases}$$



Quindi si ricavano le reazioni vincolari dalle 2.7, le quali porgono


$$11.6) \begin{cases} R_x^i = -mAG (\ddot{\theta}_i \sin \theta_i + \dot{\theta}_i^2 \cos \theta_i) \\ R_y^i = mg + mAG (\ddot{\theta}_i \cos \theta_i - \dot{\theta}_i^2 \sin \theta_i) \end{cases}$$











12. Algoritmo risolutivo. Per effettuare l'integrazione numerica sono stati scritti in Fortran i seguenti codici:


- l'unità chiamante *main_ese_14_a*, la quale si occupa del calcolo della legge oraria e delle reazioni vincolari;
- il modulo *mod_ese_14_a*, il quale contiene le costanti geometriche 1.1 oltre alle seguenti procedure di modulo:

- subroutine *diagramma*, la quale si occupa di tracciare l'andamento nel tempo delle funzioni $\theta(t), \dot{\theta}(t), \ddot{\theta}(t)$;
- subroutine *reazioni*, la quale si occupa di tracciare l'andamento nel tempo delle funzioni $R_x(t), R_y(t)$;
- subroutine *pendolo*, la quale si occupa di tracciare i fotogrammi con la posizione del pendolo (si tratta del pendolo equivalente circolare di cui si parla nel paragrafo iniziale) per più istanti consecutivi; questi fotogrammi sono stati poi usati per realizzare una animazione della oscillazione del pendolo.

In figura il diagramma di flusso complessivo.

13. Esecuzione del programma. Il programma restituisce l'andamento nel tempo delle seguenti funzioni, che riporto nelle figure precedenti e nella successiva.

- $\theta = \theta(t), \dot{\theta} = \dot{\theta}(t), \ddot{\theta} = \ddot{\theta}(t)$
- $r_x = r_x(t), r_y = r_y(t), r = r(t)$



14. Codice del programma chiamante. Segue lo script della unità chiamante del programma che risolve il quesito.

```
PROGRAM main_ese_14_a

USE DISLIN    !libreria grafica
USE mod_ese_14_a

!sezione dichiarativa

IMPLICIT NONE
```

```

!dichiaro gli array di theta e delle sue derivate
REAL,DIMENSION(0:n-1):: theta, theta_p, theta_p_p

!dichiaro gli array delle componenti e del valore totale della reazione vincolare
REAL,DIMENSION(0:n-1):: rx, ry, r

!dichiaro l'array del tempo
REAL, DIMENSION(0:n-1):: tempo

!dichiaro gli array della posizione del baricentro
REAL,DIMENSION(0:n-1):: xG, yG

!dichiaro gli array dei coefficienti k del metodo Runge-Kutta
REAL,DIMENSION(0:n-1):: k1, k2, k3, k4

!dichiaro la costante A presente nella equazione differenziale
REAL:: A

!dichiaro gli indici del ciclo e una variabile di lavoro
INTEGER:: i,j,l,p,k

!indico una stringa che uso per chiudere il programma
CHARACTER(len=10):: chiusura

!dichiaro i titoli dei diagrammi
CHARACTER(len=13):: titolo_1 = "theta  "
CHARACTER(len=13):: titolo_2 = "theta primo "
CHARACTER(len=13):: titolo_3 = "theta secondo"
CHARACTER(len=13):: unita_1 = "radianti  "
CHARACTER(len=13):: unita_2 = "radianti/s "
CHARACTER(len=13):: unita_3 = "radianti/s^2 "

!sezione esecutiva

!calcolo h
h = delta_t/REAL(n-1)

WRITE(*,*) h

!calcolo la costante A
A = SQRT( ( (m*AG**2.)+IG )/( 2.*AG*m*g ) )

!impongo la condizione iniziale
theta(0) = 3.141593
tempo(0) = 0.

!ciclo del metodo di Runge-Kutta

```

andata_1: DO i=0, n, 1

IF (theta(i)>=6.283) EXIT andata_1

!calcolo la velocità angolare e l'accelerazione angolare

theta_p(i) = (1./A)*SQRT((-1.)*SIN(theta(i)))
theta_p_p(i) = (-1.)*COS(theta(i))/(2*A**2.)

!calcolo la posizione del baricentro

xG(i) = AG*COS(theta(i))
yG(i) = AG*SIN(theta(i))

!calcolo le componenti della reazione vincolari e il modulo della reazione

rx(i) = (-1.)*m*AG*(theta_p_p(i)*SIN(theta(i)) + (theta_p(i)**2)*COS(theta(i)))
ry(i) = m*g + m*AG*(theta_p_p(i)*COS(theta(i)) - (theta_p(i)**2)*SIN(theta(i)))
r (i) = SQRT(rx(i)**2 + ry(i)**2)

!calcolo i coefficienti k del metodo runge-kutta

k1(i) = (1./A)*SQRT((-1.)*SIN(theta(i)))
k2(i) = (1./A)*SQRT((-1.)*SIN(theta(i) + 0.5*h*k1(i)))
k3(i) = (1./A)*SQRT((-1.)*SIN(theta(i) + 0.5*h*k2(i)))
k4(i) = (1./A)*SQRT((-1.)*SIN(theta(i) + h*k3(i)))

!calcolo il valore di theta(i+1)

theta(i+1) = theta(i) + (h/6.)*(k1(i) + 2*k2(i) + 2*k3(i) + k4(i))

!calcolo i valori dell'array del tempo

tempo(i+1) = tempo(i) + h

!assegno il valore k

k=i

WRITE(*,*) theta(i), rx(i), ry(i), r(i)

END DO andata_1

!assegno gli altrivalori degli array invertendo il moto sin qui ottenuto

ritorno_1: DO j=1, k+1, 1

theta(k+j) = theta(k-(j-1))
theta_p(k+j) = theta_p(k-(j-1))
theta_p_p(k+j) = theta_p_p(k-(j-1))

xG(k+j) = AG*COS(theta(k+j))
yG(k+j) = AG*SIN(theta(k+j))

rx(k+j)= rx(k-(j-1))
ry(k+j)= ry(k-(j-1))
r (k+j) = r(k-(j-1))

!calcolo i valori dell'array del tempo

tempo(k+j) = tempo(k+j-1) + h

WRITE(*,*) theta(k+j), rx(k+j), ry(k+j), r(k+j)

END DO ritorno_1

!assegno i valori per la seconda andata del pendolo

andata_2: DO l=2, k, 1

theta(k+k+l) = theta(l)
theta_p(k+k+l) = theta_p(l)
theta_p_p(k+k+l) = theta_p_p(l)

xG(k+k+l) = AG*COS(theta(k+k+l))
yG(k+k+l) = AG*SIN(theta(k+k+l))

rx(k+k+l)=rx(l)
ry(k+k+l)=ry(l)
r(k+k+l) =r (l)

!calcolo i valori dell'array del tempo

tempo(k+k+l) = tempo(k+k+l-1) + h

WRITE(*,*) theta(k+k+l), rx(k+k+l), ry(k+k+l), r(k+k+l)

END DO andata_2

!assegno i valori del secondo ritorno

ritorno_2: DO p=1, k+1, 1

theta(k+k+k+p) = theta(k-(p-1))
theta_p(k+k+k+p) = theta_p(k-(p-1))
theta_p_p(k+k+k+p) = theta_p_p(k-(p-1))

xG(k+k+k+p) = AG*COS(theta(k+k+k+p))
yG(k+k+k+p) = AG*SIN(theta(k+k+k+p))

rx(k+k+k+p)=rx(k-(p-1))
ry(k+k+k+p)=ry(k-(p-1))
r(k+k+k+p)=r(k-(p-1))

!calcolo i valori dell'array del tempo

tempo(k+k+k+p) = tempo(k+k+k+p-1) + h

WRITE(*,*) theta(k+k+k+p), rx(k+k+k+p), ry(k+k+k+p), r(k+k+k+p)

END DO ritorno_2

WRITE(*,*) k

!traccio il diagramma di theta in funzione del tempo

CALL diagramma (theta, tempo, titolo_1, unita_1, k)

!traccio il diagramma di theta_p in funzione del tempo

```

CALL diagramma (theta_p, tempo, titolo_2, unita_2, k)

!traccio il diagramma di theta_p_p in funzione del tempo

CALL diagramma (theta_p_p, tempo, titolo_3, unita_3, k)

!traccio il diagramma della reazione vincolare in funzione del tempo

CALL reazione (rx, ry, r, tempo, k)

!traccio la posizione del pendolo istante per istante

animazione: DO i=0,k+k+k+k+1,10

                CALL pendolo (xG, yG, rx, ry, tempo, i)

END DO animazione

WRITE (*,*)"Per chiudere il programma premi una lettera qualunque."
WRITE (*,*)"Tutti i dati andranno persi."
READ (*,*) chiusura

STOP

END PROGRAM main_ese_14_a

```

15. Codice del modulo. Seguono le linee di programma del modulo utilizzato dalla unità chiamante.

```

MODULE mod_ese_14_a

!sezione dichiarativa

IMPLICIT NONE

!dichiaro i parametri inerziali e geometrici del pendolo

REAL, PARAMETER:: m = 5.0  !la massa in kg
REAL, PARAMETER:: IG = 0.05 !il momento d'inerzia in kg per m^2
REAL, PARAMETER:: AG = 0.05 !distanza fra cerniera e baricentro in m
REAL, PARAMETER:: g = 9.807 !accelerazione gravitazionale
REAL, PARAMETER:: Ra= 0.142 !raggio di un pendolo equivalente

!dichiaro il numero di iterazioni, l'intervallo e l'incremento temporale

INTEGER, PARAMETER:: n    = 10000
REAL, PARAMETER  :: delta_t = 4.
REAL              :: h

!scrivo le subroutine

CONTAINS

!-----

SUBROUTINE diagramma (funzione, tempo, stringa, unita, k)

!sezione dichiarativa

```

!dichiaro gli argomenti fittizi

!dichiaro l'array della funzione e quello del tempo

REAL,INTENT(IN),DIMENSION(0:n-1):: funzione, tempo

!dichiaro la stringa usata per la leggenda

CHARACTER(len=13),INTENT(IN):: stringa

CHARACTER(len=13),INTENT(IN):: unita

!dichiaro il numero di elementi scitti sugli array

INTEGER, INTENT(IN):: k

!dichiaro le variabili locali

!dichiaro il valore massimo e minimo della funzione

REAL:: max, min

!sezione esecutiva

!calcolo il valore massimo della funzione

max = MAXVAL (funzione(0:k+1))

min = MINVAL (funzione(0:k+1))

!imposto il formato del file

CALL METAFL ('bmp') !indico il formato dell'output

CALL BMPMOD (25000,'meter','resolution') !risoluzione del formato .bmp

!imposto la pagina

CALL SCRMOD ('revers') !scritta nera su fondo bianco

CALL DISINI !richiama alcune impostazioni di default

CALL PAGERA !traccio un bordo per il piano xy

CALL DUPLX !font a doppio spessore

!imposto gli assi x,y

CALL AXSPOS (700,2700) !coordinate angolo basso sinistra

CALL AXSLEN (1700,1300) !lunghezza dei due assi in pixel

CALL NAME ('secondi','x') !nome delle ascisse

CALL NAME (unita,'y') !nome delle ordinate

CALL LABDIG (3,'x') !chiedo 3 cifre decimali per l'asse x

CALL DASH !tratteggio per gli assi coordinati

!inizio, fine, intervallo assi x, y

CALL GRAF (0.0,h*(k+k+k+k+1),0.0,delta_t/10,min-0.5,max+0.5,min,max/10.)

CALL XAXGIT !traccio la retta x=0

CALL YAxGIT !traccio la retta y=0

!traccio la curva

CALL MYLINE (1,1) !chiede una linea continua per la curva

CALL CURVE (tempo, funzione, k+k+k+k+1) !traccio lo spazio in funzione del tempo

```

!imposto il titolo

CALL TITLIN (stringa,1)    !prima riga del titolo
CALL TITLE                !stampa il titolo di cui sopra

CALL DISFIN

END SUBROUTINE diagramma

!-----

SUBROUTINE reazione (rx, ry, r, tempo, k)

!sezione dichiarativa

!dichiaro gli argomenti fittizi

!dichiaro l'array della funzione e quello del tempo
REAL,INTENT(IN),DIMENSION(0:n-1):: rx, ry, r, tempo

!dichiaro il numero di elementi scitti sugli array

INTEGER, INTENT(IN):: k

!dichiaro le variabili locali

!dichiaro la stringa usata per la leggenda

CHARACTER(len=13):: stringa

!dichiaro i valori massimi e minimi delle funzioni rx, ry, r

REAL:: max_rx, max_ry, max_r
REAL:: min_rx, min_ry, min_r

!dichiaro il valore massimo dei massimi e minimo dei minimi

REAL:: max_t
REAL:: min_t

!sezione esecutiva

!calcolo il valore massimo e minimo delle funzioni

max_rx = MAXVAL (rx(0:k+1))
min_rx = MINVAL (rx(0:k+1))
max_ry = MAXVAL (ry(0:k+1))
min_ry = MINVAL (ry(0:k+1))
max_r  = MAXVAL (r(0:k+1) )
min_r  = MINVAL (r(0:k+1) )

!calcolo il valore massimo dei massimi e il minimo dei minimi

max_t = MAX(max_rx, max_ry, max_r)
min_t = MIN(min_rx, min_ry, min_r)

!imposto il formato del file

CALL METAFL ('bmp')          !indico il formato dell'output

```

```

CALL BMPMOD (25000,'meter','resolution') !risoluzione del formato .bmp

!imposto la pagina

CALL SCRMOD ('revers') !scritta nera su fondo bianco
CALL DISINI          !richiama alcune impostazioni di default
CALL PAGERA         !traccio un bordo per il piano xy
CALL DUPLX          !font a doppio spessore

!imposto gli assi x,y

CALL AXSPOS (900,2700) !coordinate angolo basso sinistra
CALL AXSLEN (1700,1300)!lunghezza dei due assi in pixel
CALL NAME ('secondi','x') !nome delle ascisse
CALL NAME ('newton','y') !nome delle ordinate
CALL LABDIG (3,'x') !chiedo 3 cifre decimali per l'asse x
CALL DASH          !tratteggio per gli assi coordinati

!inizio, fine, intervallo assi x, y
CALL GRAF (0.0,h*(k+k+k+k+1),0.0,delta_t/10,min_t,max_t,min_t,max_t/10)

CALL XAXGIT        !traccio la retta x=0
CALL YAxGIT        !traccio la retta y=0

!traccio la curva

CALL MYLINE (1,1)  !chiede una linea continua per la curva
CALL CHNCRV ('line') !usa una linea diversa per ogni curva
CALL CURVE (tempo, rx, k+k+k+k+1) !traccio rx in funzione del tempo
CALL CURVE (tempo, ry, k+k+k+k+1) !traccio ry in funzione del tempo
CALL CURVE (tempo, r, k+k+k+k+1) !traccio r in funzione del tempo

!imposto il titolo

CALL TITLIN ('Reazione',1) !prima riga del titolo
CALL TITLE          !stampa il titolo di cui sopra

!imposto la legenda

CALL LEGINI (stringa,3,13) !variabile di carattere, righe, lunghezza
CALL LEGLIN (stringa,'componete x',1) !prima riga della legenda
CALL LEGLIN (stringa,'componete y',2) !seconda riga della legenda
CALL LEGLIN (stringa,'modulo',3)!terza riga della legenda
CALL LEGTIT ('Legenda') !titolo legenda
CALL LEGEND (stringa,3)!posizine in alto a destra

CALL DISFIN

END SUBROUTINE reazione

!-----

SUBROUTINE pendolo (xG, yG, rx, ry, tempo, i)

!sezione dichiarativa

!dichiaro gli argomenti fittizi

!dichiaro gli array delle coordinate di G e del tempo

```

```

REAL,INTENT(IN),DIMENSION(0:n-1):: xG, yG, rx, ry, tempo

!dichiaro il numero del fotogramma

INTEGER, INTENT(IN):: i

!dichiaro le variabili locali

!stringa usata per la legenda

CHARACTER(len=30)::stringa

!sezione esecutiva

!imposto il tipo di file

CALL METAFL ('bmp') !indico il formato dell'output
CALL BMPMOD (25000,'meter','resolution')!risoluzioneformato .bmp

!imposto la pagina

CALL SCRMOD ('revers') !scritta nera su fondo bianco
CALL DISINI          !richiama alcune impostazioni di default
CALL PAGERA         !traccio un bordo per il piano xy
CALL DUPLX          !font a doppio spessore
CALL MYLINE (1,1)   !linea continua

!imposto gli assi

CALL AXSPOS (700,2000)
!coordinate angolo basso sinistra
CALL AXSLEN (1700,1700)
!lunghezza dei due assi in pixel
CALL NAME ('metri','x')      !nome delle ascisse
CALL NAME ('metri','y')      !nome delle ordinate
CALL GRAF (-0.196,0.196,-0.196,0.05,-0.196,0.145,-0.196,.05)
CALL XAXGIT
                                !traccio la retta x=0
CALL YAxGIT
                                !traccio la retta y=0

CALL DASH
!tratteggio per gli assi coordinati
CALL NAME ('metri','x')
!nome delle ascisse
CALL NAME ('metri','y')
!nome delle ordinate

!imposto il titolo

CALL TITLIN ("Pendolo",1)
!prima riga del titolo
CALL TITLE          !stampa il titolo di cui sopra

!disegno il pendolo e la reazione vincolare istante per istante

!traccio la circonferenza del pendolo

CALL RLCIRC (xG(i), yG(i), Ra)

!traccio la cerniera

```

```
CALL RLCIRC (0.,0.,0.01)
```

```
!traccio il vettore che rappresenta la forza peso
```

```
CALL RLVEC (xG(i), yG(i), xG(i), yG(i)-m*g/1000.0, 2211)
```

```
!traccio il vettore che rappresenta la reazione vincolare
```

```
CALL RLVEC (0., 0., rx(i)/1000.0, ry(i)/1000.0, 2211)
```

```
CALL DISFIN
```

```
END SUBROUTINE pendolo
```

```
!-----
```

```
END MODULE mod_ese_14_a
```